//Fox_FinancialDimensions - created on 19 Feb 2017 by Admin BEGIN
static void Fox_DimensionValuesFromDefaultDimension(Args _args)
{
    #define.MainAccountId('110170')
    #define.Company('usmf')
    #define.ChartOfAccount('shared')

    CompanyInfo                         companyInfo;
    Counter                             counter;
    Common                              common;
    DictTable                           dictTable;
    DimensionAttributeValueSetStorage   dimensionValueStorageClass;
    DimensionAttribute                  dimensionAttribute;
    DimensionAttributeValueFinancialStmt dimensionAttributeFin;
    DimensionAttributeValue             dimensionAttributeValue;
    LedgerChartOfAccounts               ledger;
    MainAccount                         mainAccount;
    MainAccountLegalEntity              mainAccountLegalEntity;
    str                                 name;
    
    select firstOnly DefaultDimension
        from mainAccountLegalEntity
    exists  join companyInfo
        where companyInfo.RecId         == MainAccountLegalEntity.LegalEntity
           && companyInfo.DataArea      == #Company
    exists join mainAccount
        where mainAccount.RecId         == MainAccountLegalEntity.MainAccount
           && MainAccount.MainAccountId == #MainAccountId
    exists join ledger
        where ledger.RecId              == mainAccount.LedgerChartOfAccounts
           && ledger.Name               == #ChartOfAccount;

    dimensionValueStorageClass =     	DimensionAttributeValueSetStorage::find(mainAccountLegalEntity.DefaultDimension);

    for (counter = 1 ; counter <= dimensionValueStorageClass.elements() ; counter++)
    {
        select firstonly EntityInstance from dimensionAttributeValue
            where dimensionAttributeValue.RecId == dimensionValueStorageClass.getValueByIndex(counter)
        join BackingEntityType, KeyAttribute, Name, NameAttribute, RecId from dimensionAttribute
            where dimensionAttribute.RecId == dimensionAttributeValue.DimensionAttribute;

        if (dimensionAttribute && dimensionAttributeValue)
        {
            dictTable = new DictTable(dimensionAttribute.BackingEntityType);
            common    = dictTable.makeRecord();

            if (common.TableId)
            {
                select firstOnly common
                    where common.(dimensionAttribute.KeyAttribute) == 								dimensionAttributeValue.EntityInstance;

                name = common.(dimensionAttribute.NameAttribute);
            }

            info(strfmt("%1  -- %2 -- %3 ", dimensionAttribute.name,              														dimensionValueStorageClass.getDisplayValueByDimensionAttribute(dimensionAttribute.RecId),
                    name
                     ));
        }
    }
}

